Sample solutions of selected problems from Sets 10 and 11

Chris Peterson and Hirotachi Abo

All the computations will be done over the rationals $\mathbb{Q}$. Let $k =$`QQ`.

```
i1 : KK=QQ;
```

Problem 2 (Set 10). Let $I = (x^3y - x^2, x^2 - xy)$ be an ideal in $k[x, y]$. We compute the ideal quotient $I : x^2$ by taking the following steps:

(1) Compute the intersection of the ideal I and (x^2);

(2) Let $(h_1, \ldots, h_t)$ be the intersection of these two ideal obtained in (1). Compute $(h_1/x^2, \ldots, h_t/x^2)$.

The result in (2) equals the ideal quotient $I : x^2$.

(1) Let L be the ideal in $k[t, x, y]$ defined by $tI + (1 - t)(x^2)$. Recall that $L \cap k[x, y]$ is the intersection of I and (x^2). To compute $L \cap k[x, y]$, we use elimination:

```
i2 : S=KK[x,y];

i3 : I=ideal(x^3*y-x^2,x^2-x*y);

o3 : Ideal of S

i4 : J=ideal(x^2);

o4 : Ideal of S

i5 : R=KK[t,x,y,MonomialOrder=>Eliminate 1];

i6 : L=t*substitute(I,R)+(1-t)*substitute(J,R);

o6 : Ideal of R

i7 : elimL=ideal selectInSubring(1,gens gb L)

                3     2     2 2     2
o7 = ideal (x  - x y, x y  - x )
```

```
o7 : Ideal of R
```

Substitute `elimL` into the ring `S`:

```
i8 : use S

o8 = S

o8 : PolynomialRing

i9 : K=substitute(elimL,S);

o9 : Ideal of S
```

Output `o9` is the intersection of I and (x^2). Divide each generator of `K'` by x^2:

```
i10 : K'=substitute(ideal((gens K)_(0,0)/x^2,(gens K)_(0,1)/x^2),S)

                            2
o10 = ideal (x - y, y  - 1)

o10 : Ideal of S
```

Compute the ideal quotient $I : x^2$ with `I:J` and compare the results:

```
i11 : K'==(I:J)

o11 = true
```

Problem 4 (Set 11). Define the coordinate ring $k[w, x, y, z]$ of $\mathbb{A}^4$ and the ideal $I = (w^2 - x, w^3 - y, w^4 - z)$ in $k[w, x, y, z]$:

```
i12 : S=KK[w..z,MonomialOrder=>GLex]

o12 = S

o12 : PolynomialRing

i13 : I=ideal(w^2-x,w^3-y,w^4-z)
```

```
             2       3       4
o13 = ideal (w  - x, w  - y, w  - z)

o13 : Ideal of S
```

Define the new ring `Sh`= $k[w, x, z, y, h]$. Let `J` be the ideal in `Sh` obtained from `I` by homogenizing the given generators of `I`. This ideal can be given with the command `homogenize`:

```
i14 : Sh=KK[h,w..z]

o14 = Sh

o14 : PolynomialRing

i15 : I'=substitute(I,Sh)

             2       3       4
o15 = ideal (w  - x, w  - y, w  - z)

o15 : Ideal of Sh

i16 : J=homogenize(I',h)

             2         3    2    4    3
o16 = ideal (w  - h*x, w  - h y, w  - h z)

o16 : Ideal of Sh
```

(a) Saturate the ideal `J` with respect to the homogenizing variable:

```
i17 : L=saturate(J,h)

             2                                2                   2
o17 = ideal (y  - x*z, x*y - w*z, w*y - h*z, x  - h*z, w*x - h*y, w  - h*x)

o17 : Ideal of Sh
```

The ideal `L` is the ideal of the projective closure of $V(I)$.

(b) and (c) There is another way to find the ideal of the projective closure of $V(I)$. Compute the gröbner basis `stdI` for `I` with respect to the homogeneous lexicographic order and then homogenize the elements in `stdI`:

```
i18 : stdI=ideal gens gb I

                       2   2                             2            2         2    2
o18 = ideal (x*z - y , x  - z, w*z - x*y, w*y - x , w*x - y, w  - x, x*y  - z ,

o18 : Ideal of S

i19 : stdI'=substitute(stdI,Sh)

                   2         2                        2                  2         2
o19 = ideal (- y  + x*z, x  - z, - x*y + w*z, - x  + w*y, w*x - y, w  - x, x*y

o19 : Ideal of Sh

i20 : L'=homogenize(stdI',h)

                   2         2                          2                  2
o20 = ideal (- y  + x*z, x  - h*z, - x*y + w*z, - x  + w*y, w*x - h*y, w  - h*x,

o20 : Ideal of Sh
```

Are `L` and `L'` the same? `Macaulay2` can answer to this question:

```
i21 : L==L'

o21 = true
```

Remark. Why can `Macaulay2` say that two given ideals I and J are the same? This is based on the following fact: Let I be an ideal in a polynomial ring. Given a monomial order, I has a unique reduced gröbner basis. `Macaulay2` computes the reduced gröbner bases of I and J and compares these gröbner bases. If I and J have the same reduced gröbner basis, then $I = J$.

Problem 5 (Set 11). Define the polynomial ring `S`$=k[x,y,z]$ with homogeneous lexicographic order and the ideal $I = (x^2 - y, x^3 - z)$ in `S`:

```
i22 : S=KK[x..z,MonomialOrder=>GLex]

o22 = S

o22 : PolynomialRing

i23 : I=ideal(x^2-y,x^3-z)
```

```
                2       3
o23 = ideal (x  - y, x  - z)

o23 : Ideal of S
```

Let Sh be the polynomial ring $k[x,y,z,h]$. Compute the homogeneization I^h of I in the same way as in Problem 4:

```
i24 : stdI=ideal gens gb I

                     2            2       3    2
o24 = ideal (x*z - y , x*y - z, x  - y, y  - z )

o24 : Ideal of S

i25 : Sh=KK[h,x..z]

o25 = Sh

o25 : PolynomialRing

i26 : I'=substitute(stdI,Sh);

o26 : Ideal of Sh

i27 : Ih=homogenize(I',h)

                 2                    2          3      2
o27 = ideal (- y  + x*z, x*y - h*z, x  - h*y, y  - h*z )

o27 : Ideal of Sh
```

Consider the ideal J obtained from I by homogenizing the generators of I:

```
i28 : J=homogenize(substitute(I,Sh),h)

                2         3    2
o28 = ideal (x  - h*y, x  - h z)

o28 : Ideal of Sh
```

Check that the statement $J = I^h \cap (J : I^h)$ is true:

```
i29 : J'=intersect(Ih,(J:Ih))

                2                  2
o29 = ideal (x  - h*y, h*x*y - h z)

o29 : Ideal of Sh

i30 : J'==J

o30 = true
```

(b) Recall that the radical of an ideal can be computed with `radical`:

```
i31 : radJIh=radical(J:Ih)

o31 = ideal (x, h)

o31 : Ideal of Sh
```

(c) We've shown that the radical of the intersection of two ideals equals the intersection of radicals of these ideals (see Problem 4 in Problem Set 10). Thus, from (a) it follows that the radical of J is obtained as the intersection of I^h and $(J : I^h)$:

```
i32 : radJ=intersect(Ih,radJIh)

                          2
o32 = ideal (x*y - h*z, x  - h*y)

o32 : Ideal of Sh

i33 : radJ==radical J

o33 = true
```

Problem 6 (Sec 11). (a) Our task is to find the ideal of the projective closure $\bar{V}$ for the image V of the polynomial map $\phi : \mathbb{A}^2 \to \mathbb{A}^5$ defined by $\phi(x, y) = (x, y, x^2, xy, y^2)$. First of all, we find the ideal of V. Clear the earlier meaning of `w` to make it into a subscripted variable:

```
i34 : erase global w

o34 = w

o34 : Symbol
```

Define the coordinate rings of $\mathbb{A}^5$ and $\mathbb{A}^2$:

```
i35 : ringA5=KK[w_1..w_5]

o35 = ringA5

o35 : PolynomialRing

i36 : ringA2=KK[x,y]

o36 = ringA2

o36 : PolynomialRing
```

The ideal $I(V)$ can be obtained as the kernel of the corresponding ring homomorphism from `ringA5` to `ringA2`:

```
i37 : I=ker map(ringA2,ringA5,{x,y,x^2,x*y,y^2})

               2                   2                                   2
o37 = ideal (w  - w , w w  - w , w  - w , w w  - w w , w w  - w w , w  - w w )
              2    5   1 2    4   1    3   2 4    1 5   2 3    1 4   4    3 5

o37 : Ideal of ringA5
```

Let `ringP5` be the homogeneous coordinate ring $k[w_0, w_1, \ldots, w_5]$ of $\mathbb{P}^5$:

```
i38 : ringP5=KK[w_0..w_5]

o38 = ringP5

o38 : PolynomialRing
```

Identify $\mathbb{A}^5$ with the open subset of $\mathbb{P}^5$ defined by $\mathbb{P}^5 \setminus V(w_0)$. The homogeneous ideal of $\bar{V}$ is therefore obtained by computing the ideal quotient $I : w_0^\infty$:

```
i39 : I'=substitute(I,ringP5)

              2                   2                                  2
o39 = ideal (w  - w , w w  - w , w  - w , w w  - w w , w w  - w w , w  - w w )
              2    5   1 2    4   1    3   2 4    1 5   2 3    1 4   4    3 5

o39 : Ideal of ringP5

i40 : J=saturate(homogenize(I',w_0),w_0)

              2                                    2                       2
o40 = ideal (w  - w w , w w  - w w , w w  - w w , w  - w w , w w  - w w , w  - w
              4    3 5   2 4    1 5   2 3    1 4   2    0 5   1 2    0 4   1

o40 : Ideal of ringP5
```

(b) Define the polynomial map $\phi : \mathbb{P}^2 \to \mathbb{P}^5$ by $\phi([x : y : z]) = [x^2 : xy : xz : y^2 : yz : z^2]$. This map induces a ring homomorphism $\tilde{\phi} : k[w_0, \dots, w_5] \to k[x, y, z]$. Compute the kernel of this homomorphism:

```
i41 : ringP2=KK[x,y,z]

o41 = ringP2

o41 : PolynomialRing

i42 : K=ker map(ringP2,ringP5,{x^2,x*y,x*z,y^2,y*z,z^2})

              2                                    2                       2
o42 = ideal (w  - w w , w w  - w w , w w  - w w , w  - w w , w w  - w w , w  - w
              4    3 5   2 4    1 5   2 3    1 4   2    0 5   1 2    0 4   1

o42 : Ideal of ringP5
```

The kernel is the desired ideal. Check that `J` and `K` are the same:

```
i43 : J==K

o43 = true
```

Problem 7 (Set 11). Let `ringP2` be the homogeneous coordinate ring $k[y_0, y_1, y_2]$ of $\mathbb{P}^2$ and let `ringP9` be the homogeneous coordinate ring $k[x_0, \dots, x_9]$ of $\mathbb{P}^9$.

Consider the polynomial map $\phi : \mathbb{P}^2 \to \mathbb{P}^9$ defined by

$$\phi([y_0 : y_1 : y_1]) = [y_0^3 : y_0^2 y_1 : y_0^2 y_2 : y_0 y_1^2 : y_0 y_1 y_2 : y_0 y_2^2 : y_1^3 : y_1^2 y_2 : y_1 y_2^2 : y_2^3].$$

The ideal corresponding the image of this map can be computed in the same way as in the previous problem. This map can be defined by using the command `basis`:

```
i44 : erase global x

o44 = x

o44 : Symbol

i45 : erase global y

o45 = y

o45 : Symbol

i46 : ringP9=KK[x_0..x_9]

o46 = ringP9

o46 : PolynomialRing

i47 : ringP2=KK[y_0..y_2]

o47 = ringP2

o47 : PolynomialRing

i48 : I=ker map(ringP2,ringP9,basis(3,ringP2))

               2
o48 = ideal (x   - x x , x x  - x x , x x  - x x , x x  - x x , x x  - x x , x x
               8    7 9   5 8    4 9   7 8    6 9   4 8    3 9   2 8    1 9   5 7

o48 : Ideal of ringP9
```

Remark. The command `basis` takes as inputs an integer l and a graded ring $R = \bigoplus_{d \in \mathbb{Z}} R_d$ (or more generally a graded module) and returns the dimension of R_l.